

TEXAS A&M UNIVERSITY  
DEPARTMENT OF INDUSTRIAL & SYSTEMS ENGINEERING

# AMPL/CPLEX PROJECT

---

ISEN 622 LINEAR PROGRAMMING

Submitted by:

SEYEDMOHAMMADHOSSEIN HOSSEINIAN

622000417

ALLU JAYA BHARATH

524001234

HUA-YU YUAN

124002665

Fall 2015

## TABLE OF CONTENTS

---

1	Problem Description .....	1
1.1	OBJECTIVE .....	2
1.2	ASSUMPTIONS .....	2
2	Mathematical Formulation .....	3
3	AMPL Model File .....	5
4	AMPL Data File .....	7
5	AMPL Run File (basic).....	8
6	Solution (basic).....	9
6.1	AMPL LOG FILE (OUTPUT) .....	9
6.2	RESULTS (BASIC RECOMMENDATION) .....	11
7	Question Part a .....	12
7.1	RECOMMENDATION .....	13
7.2	AMPL RUN FILE.....	14
7.3	AMPL LOG FILE (OUTPUT) .....	15
8	Question Part b .....	17
8.1	RECOMMENDATION .....	17
8.2	AMPL RUN FILE.....	19
8.3	AMPL LOG FILE (OUTPUT) .....	20
9	Question Part c.....	22
10	Question Part d .....	23
10.1	AMPL LOG FILE (OUTPUT) .....	24

# 1 PROBLEM DESCRIPTION

## BestChip: Expansion Strategy

BestChip (BC) is a large nationwide corporation that produces low-fat snack products for an expanding market (pun intended). Basically, BC takes materials (corn, wheat, and potatoes) and turns them into two types of snacks: chips (regular and green onion) and party mix (one variety). BC is expanding into the western United States and is considering sites for locating production facilities.

BC currently has eight candidate sites. Table 20 shows the sites' purchase prices and the purchase and shipping cost per ton of each material to each site.

The purchase cost represents the yearly amortized cost of opening and operating the site (exclusive of shipping costs). Each site may produce as many as 20,000 tons of product per year.

BC has six major customers, and all demand is shipped by truck from the plant to the customer warehouse. The shipping cost depends on the tonnage and distance and comes to \$0.15 per ton-mile. The customers, their location, and their yearly demand in tons for each product are listed in Table 21. You must meet demand.

The makeup of the products does not depend on the production plant. Table 22 gives the product-ingredient mix data. The company requires that we consolidate our business, so we cannot locate plants in more than two states.

For this analysis, ignore the differences in property and income tax rates between the states (this is usually critical, but it gets us far afield of the key issue of math programming). In addition, many critical factors actually determine locations; for example, the method of financing the site purchase will also be a major factor in the decision—but we will ignore that also.

**TABLE 20**  
Site Information and Material Shipping Cost

Site Location	Purchase Cost (\$/Year)	Material Shipping Cost (\$/Ton)		
		Corn	Wheat	Potato
Yuma, AZ	125,000	10	5	16
Fresno, CA	130,000	12	8	11
Tucson, AZ	140,000	9	10	15
Pomona, CA	160,000	11	7	14
Santa Fe, NM	150,000	8	14	10
Flagstaff, AZ	170,000	10	12	11
Las Vegas, NV	155,000	13	12	9
St. George, UT	115,000	14	15	8

**TABLE 21**  
Demand Information

Company	Location	Demand		
		Regular	Green Onion	Party Mix
Jones	Salt Lake City	1,300	900	1,700
YZCO	Albuquerque	1,400	1,100	1,700
Square Q	Phoenix	1,200	800	1,800
AJ Stores	San Diego	1,900	1,200	2,200
Sun Quest	Los Angeles	1,900	1,400	2,300
Harm's Path	Tucson	1,500	1,000	1,400

**TABLE 22**  
Product-Ingredient Mix

Product	Ingredient		
	Corn	Wheat	Potato
Regular chips	70	20	10
Green onion chips	30	15	55
Party mix	20	50	30

Use Google Map to find the road distance between any two cities you need to prepare your data file.

Prescribe the plan of expansion of BestChip to the western US including all the decisions which must be made by formulating and solving a linear Mixed Binary Programming (MBP) model. Your model should be linear and should only consists of continuous or binary (0-1), variables. Also answer the following questions independently:

- If gasoline gets more expensive increasing the trucking cost for shipping the demand, then how is your recommendation affected?
- If the rail freight cost for material shipping increases, then how is your recommendation affected?
- How much time does CPLEX take to solve your linear MBP problem?
- Solve the linear programming relaxation of your model. How much time does that take? How much difference do you notice between the objective value of the LP relaxation and your original MBP? Does your LP relaxation give a feasible solution to the MBP?

---

## 1.1 OBJECTIVE

The objective of this study is to formulate and implement a linear Mixed Binary Programming (MBP) model to identify the minimum-cost expansion plan for BestChip to the western US based on the data presented. Besides, the model is used to answer the questions following the problem description.

## 1.2 ASSUMPTIONS

- As requested in the problem description, the model ignores the differences in the property and income tax rates, as well as some financial factors such as the method of financing.
- As indicated in the problem description, the makeup of the products does not depend on the production plant (i.e. site). Thus it has no effect on the optimal solution.
- The data in Table 20 is treated as the “shipping” cost of ingredients which vary by site location and ingredient type. The “purchase” cost of ingredients is assumed to be constant for all sites and so has no effect on the optimal solution. (Note that there is a discrepancy between the general description and headings of Table 20 in this regard.) This assumption is necessary to answer the question part b.
- To answer the question part b, it is assumed that the shipping cost (rail freight) of all ingredients to all sites (Table 20) increases at the same rate.
- Distances between the sites and customers are used according to the following table:

Table 1: Distances between sites and customers (miles)

		Customers					
		Jones (Salt Lake City, UT)	YZCO (Albuquerque, NM)	Square Q (Phoenix, AZ)	AJ Stores (San Diego, CA)	Sun Quest (Los Angeles, CA)	Harm's Path (Tucson, AZ)
Sites	Yuma, AZ	714	603	185	172	272	240
	Fresno, CA	816	918	594	341	222	708
	Tucson, AZ	776	449	116	410	488	0
	Pomona, CA	660	763	345	115	30	460
	Santa Fe, NM	625	65	480	832	848	510
	Flagstaff, AZ	515	325	144	490	464	260
	Las Vegas, NV	420	574	298	332	270	413
	St. George, UT	302	530	417	449	388	532

## 2 MATHEMATICAL FORMULATION

---

### Indices

$s \in \{1,2, \dots, S\}$	States
$i \in \{1,2, \dots, I\}$	Sites (plants)
$j \in \{1,2, \dots, J\}$	Customers
$m \in \{1,2, \dots, M\}$	Ingredients
$n \in \{1,2, \dots, N\}$	Products

### Parameters

$PC_i$	Purchase cost of site $i$ per year (\$/year)
$Cap_i$	Production capacity of site $i$ per year (ton/year)
$SI_{mi}$	Cost of shipping ingredient $m$ to site $i$ per year (\$/ton)
$Dis_{ij}$	Distance of site $i$ from customer $j$ (mile)
$UnitShip$	Unit cost of shipping products to customers per mile per ton (\$/mile.ton)
$Dmd_{nj}$	Demand of customer $j$ for product $n$ (ton/year)
$MR_{mn}$	Mix ratio (i.e. % of ingredient $m$ in product $n$ )
$MaxSt$	The maximum allowed number of states to have an operating site
$SS_{is}$	State-Site indicator matrix [ $SS_{is} = 1$ if site $i$ is in state $s$ , and $SS_{is} = 0$ otherwise]
$D$	A very large number [ $D \geq \max$ (sum of all demands , number of sites)]

### Variables

$X_{mi}$	Amount of ingredient $m$ shipped to site $i$ (ton/year)
$Y_{nij}$	Amount of product $n$ produced at site $i$ and shipped to customer $j$ (ton/year)
$W_i$	Site operation indicator [ $W_i = 1$ if site $i$ is operating, and $W_i = 0$ otherwise]
$Z_s$	State operation indicator [ $Z_s = 1$ if state $s$ has an operating site, and $Z_s = 0$ otherwise]

### Formulation

$$\text{Minimize} \quad \sum_{i=1}^I (PC_i * W_i) + \sum_{i=1}^I \sum_{m=1}^M (SI_{mi} * X_{mi}) + \sum_{n=1}^N \sum_{i=1}^I \sum_{j=1}^J (UnitShip * Dis_{ij} * Y_{nij})$$

Subject to:

1.  $\sum_{n=1}^N \sum_{j=1}^J Y_{nij} \leq Cap_i \quad \forall i \in \{1,2, \dots, I\}$
2.  $\sum_{i=1}^I Y_{nij} = Dmd_{nj} \quad \forall n \in \{1,2, \dots, N\} \ \& \ j \in \{1,2, \dots, J\}$
3.  $\sum_{n=1}^N (MR_{mn} * \sum_{j=1}^J Y_{nij}) \leq X_{mi} \quad \forall m \in \{1,2, \dots, M\} \ \& \ i \in \{1,2, \dots, I\}$
4.  $\sum_{n=1}^N \sum_{j=1}^J Y_{nij} \leq D * W_i \quad \forall i \in \{1,2, \dots, I\}$
5.  $Z_s \leq \sum_{i=1}^I SS_{is} * W_i \leq D * Z_s \quad \forall s \in \{1,2, \dots, S\}$
6.  $\sum_{s=1}^S Z_s \leq MaxSt$
7.  $X_{mi} \geq 0 \quad \forall m \in \{1,2, \dots, M\} \ \& \ i \in \{1,2, \dots, I\}$
8.  $Y_{nij} \geq 0 \quad \forall n \in \{1,2, \dots, N\} \ \& \ i \in \{1,2, \dots, I\} \ \& \ j \in \{1,2, \dots, J\}$
9.  $W_i \in \{0,1\} \quad \forall i \in \{1,2, \dots, I\}$
10.  $Z_s \in \{0,1\} \quad \forall s \in \{1,2, \dots, S\}$

---

Minimize      Total Cost =      Sites' purchase cost (amortized cost of opening and operating the sites) +  
Cost of shipping ingredients to the sites +  
Cost of shipping products to the customers

Subjects to:

1. Production capacity:  
Each site cannot have production more than its "production capacity".  
NOTE: The problem description assumes a single capacity for all sites. This model accepts different capacities for different sites. Clearly, the single-capacity assumption is a special case of this general formulation.
2. Meeting demand:  
The expansion plan should be devised such that demands of all customers for all products are met.
3. Availability of ingredients:  
No more than available ingredients in each site (that is, ingredients shipped to that site) can be used in production.
4. Forcing values of "site operation indicators":  
Every operating site (that is, every site that is producing at least one unit of product) must have operation indicator of one. In other words, the sites with operation indicator of zero cannot have production.
5. Forcing values of "state operation indicators":  
The states with operation indicator of one have to have at least one operating site, and ones with operation indicator of zero cannot have any operating site.
6. Logistic consideration of number of states:  
The number of states with an operating site cannot be more than the maximum allowed.
7. Ingredients shipping amounts are non-negative real-valued numbers.
8. Products shipping amounts are non-negative real-values numbers.
9. "Site operation indicators" should have binary values; 1 indicates the site is operating and 0 indicates the site is not operating.
10. "State operation indicators" should have binary values; 1 indicates the state has at least one operating site, and 0 indicates there is no operating site in that state.

### 3 AMPL MODEL FILE

---

```
set STAT ordered;           # States
set SITE ordered;           # Sites (plants)
set CUST ordered;           # Customers
set INGR ordered;           # Ingredients
set PROD ordered;           # Productions

param PC {SITE} >=0;
# PC(i) = Purchase cost of site i per year ($/year)
param Cap {SITE} >=0;
# Cap(i) = Production capacity of site i per year (ton/year)
param SI {INGR,SITE} >=0;
# SI(m,i) = Cost of shipping ingredient m to site i per year ($/ton)
param Dis {SITE,CUST} >=0;
# Dis(i,j) = Distance of site i from customer j (mile)
param UnitShip >=0;
# UnitShip = Unit cost of shipping products to customers per mile per ton ($/mile.ton)
param Dmd {PROD,CUST} >=0;
# Dmd(n,j) = Demand of customer j for product n (ton/year)
param MR {INGR,PROD} >=0;
# MR(m,n) = Mix ratio (i.e. % of ingredient m in product n)
param MaxSt >=1;
# MaxSt = The maximum allowed number of states to have an operating site
param SS {SITE,STAT} >=0;
# SS(i,s) = State-Site indicator matrix [1 if site i is in state s, and 0 otherwise]
param D >=0;
# A very large number [D >= max (sum of all demands , number of sites)]

var X {INGR,SITE} >=0;
# X(m,i) = Amount of ingredient m shipped to site i (ton/year)
var Y {PROD,SITE,CUST} >=0;
# Y(n,i,j) = Amount of product n produced at site i & shipped to customer j (ton/year)
var W {SITE} binary;
# W(i) = Site operation indicator [1 if site i is operating, and 0 otherwise]
var Z {STAT} binary;
# Z(s) = State operation indicator [1 if state s has an operating site, and 0 otherwise]

minimize Total_Cost:
sum {i in SITE} PC[i]*W[i] +
sum {m in INGR , i in SITE} SI[m,i]*X[m,i] +
sum {n in PROD , i in SITE , j in CUST} UnitShip*Dis[i,j]*Y[n,i,j];
# Cost of expanding BestChip to the western US
```

---

```

subject to Capacity {i in SITE}: sum {n in PROD , j in CUST} Y[n,i,j] <= Cap [i];
# Sites (plants) cannot produce more than their capacities.

subject to Demand {n in PROD , j in CUST}: sum {i in SITE} Y[n,i,j] = Dmd[n,j];
# Sum of products from all sites to every customer should satisfy its demand.

subject to Availability {m in INGR , i in SITE}: sum {n in PROD} (MR[m,n]*sum {j in CUST} Y[n,i,j]) <= X[m,i];
# Production at every site is restricted to available ingredients in that site.

subject to ForceSITE {i in SITE}: sum {n in PROD , j in CUST} Y[n,i,j] <= D*W[i];
# The sites with "operation indicator" of zero cannot to have production.

subject to ForceSTATElb {s in STAT}: Z[s] <= sum {i in SITE} SS[i,s]*W[i];
# The states with operation indicator of one have to have at least one operating site

subject to ForceSTATEub {s in STAT}: sum {i in SITE} SS[i,s]*W[i] <= D*Z[s];
# The states with operation indicator of zero cannot have any operating site.

subject to NumStates: sum {s in STAT} Z[s] <= MaxSt;
# The number of states with an operating site cannot be more than the maximum
allowed.

```



## 4 AMPL DATA FILE

---

```
set STAT := AZ CA NM NV UT;
set SITE := Yuma Fresno Tucson Pomona SantaFe FlagStaff LasVegas StGeorge;
set CUST := Jones YZCO SquareQ AJStores SunQuest HarmsPath;
set INGR := CORN WHEAT POTATO;
set PROD := Regular GrOnion PartyMix;

param PC := Yuma 125e3 Fresno 130e3 Tucson 140e3 Pomona 160e3 SantaFe 150e3 FlagStaff
170e3 LasVegas 155e3 StGeorge 115e3;

param Cap := Yuma 20e3 Fresno 20e3 Tucson 20e3 Pomona 20e3 SantaFe 20e3 FlagStaff
20e3 LasVegas 20e3 StGeorge 20e3;

param SI: Yuma Fresno Tucson Pomona SantaFe FlagStaff LasVegas StGeorge :=
    CORN      10      12      9      11      8      10      13      14
    WHEAT      5       8      10      7      14      12      12      15
    POTATO     16      11      15      14      10      11      9       8;

param Dis: Jones YZCO SquareQ AJStores SunQuest HarmsPath :=
    Yuma      714      603      185      172      272      240
    Fresno     816      918      594      341      222      708
    Tucson     776      449      116      410      488      0
    Pomona     660      763      345      115      30       460
    SantaFe    625      65       480      832      848      510
    FlagStaff  515      325      144      490      464      260
    LasVegas   420      574      298      332      270      413
    StGeorge   302      530      417      449      388      532;

param UnitShip := 0.15;

param Dmd: Jones YZCO SquareQ AJStores SunQuest HarmsPath :=
    Regular    1300     1400     1200     1900     1900     1500
    GrOnion     900     1100     800      1200     1400     1000
    PartyMix    1700     1700     1800     2200     2300     1400;

param MR: Regular GrOnion PartyMix :=
    CORN        0.70     0.30     0.20
    WHEAT        0.20     0.15     0.50
    POTATO       0.10     0.55     0.30;

param MaxSt := 2;

param SS: AZ CA NM NV UT :=
    Yuma        1       0       0       0       0
    Fresno       0       1       0       0       0
    Tucson       1       0       0       0       0
    Pomona       0       1       0       0       0
    SantaFe      0       0       1       0       0
    FlagStaff    1       0       0       0       0
    LasVegas     0       0       0       1       0
    StGeorge     0       0       0       0       1;

param D := 1e6;
```

---

## 5 AMPL RUN FILE (BASIC)

---

```
option solver cplex;
reset;
model basic.mod;
data basic.dat;
solve;
option display_eps 0.0001;
printf "\n" >> 'basic.log';
printf "----- OPTIMAL OBJECTIVE VALUE -----\n\n" >> 'basic.log';
printf "The minimum cost of expansion is: ($/year)\n" >> 'basic.log';
display Total_Cost >> 'basic.log';
printf "\n" >> 'basic.log';
printf "----- OPTIMAL SOLUTION -----\n\n" >> 'basic.log';
printf "Here is details of the minimum-cost expansion plan:\n\n" >> 'basic.log';
printf "(i) Operating sites: " >> 'basic.log';
printf "(1 indicates the site will be operating)\n" >> 'basic.log';
display W >> 'basic.log';
printf "\n" >> 'basic.log';
printf "(ii) Amount of ingredients to be shipped to each site (ton/year):\n" >>
'basic.log';
display X >> 'basic.log';
param Production {SITE,PROD};
let {i in SITE,n in PROD} Production[i,n] := sum {j in CUST} Y[n,i,j];
printf "\n" >> 'basic.log';
printf "(iii) Production plan for each site (ton/year):\n" >> 'basic.log';
display Production >> 'basic.log';
printf "\n" >> 'basic.log';
printf "(iv) Distribution of products among the customers (ton/year):\n" >>
'basic.log';
display Y >> 'basic.log';
printf "\n" >> 'basic.log';
printf "(v) States with operating site:\n" >> 'basic.log';
printf "(1 indicates the state will have at least one operating site)\n" >>
'basic.log';
display Z >> 'basic.log';
printf "\n" >> 'basic.log';
printf "-----\n" >> 'basic.log';
quit;
```

## 6 SOLUTION (BASIC)

---

### 6.1 AMPL LOG FILE (OUTPUT)

----- OPTIMAL OBJECTIVE VALUE -----

The minimum cost of expansion is: (\$/year)  
Total\_Cost = 1440480

----- OPTIMAL SOLUTION -----

Here is details of the minimum-cost expansion plan:

(i) Operating sites: (1 indicates the site will be operating)

W [\*] :=

Yuma	0
Fresno	0
Tucson	1
Pomona	1
SantaFe	0
FlagStaff	0
LasVegas	0
StGeorge	0

;

(ii) Amount of ingredients to be shipped to each site (ton/year):

X [\*,\*] (tr)

:	CORN	WHEAT	POTATO	:=
Yuma	0	0	0	
Fresno	0	0	0	
Tucson	4720	3705	3475	
Pomona	5860	4645	4295	
SantaFe	0	0	0	
FlagStaff	0	0	0	
LasVegas	0	0	0	
StGeorge	0	0	0	

;

(iii) Production plan for each site (ton/year):

Production [\*,\*]

:	Regular	GrOnion	PartyMix	:=
Yuma	0	0	0	
Fresno	0	0	0	
Tucson	4100	2900	4900	
Pomona	5100	3500	6200	
SantaFe	0	0	0	
FlagStaff	0	0	0	
LasVegas	0	0	0	
StGeorge	0	0	0	

;

---

(iv) Distribution of products among the customers (ton/year):

Y [Regular,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	1400	1200	0	0	1500	
Pomona	1300	0	0	1900	1900	0	
SantaFe	0	0	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

[GrOnion,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	1100	800	0	0	1000	
Pomona	900	0	0	1200	1400	0	
SantaFe	0	0	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

[PartyMix,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	1700	1800	0	0	1400	
Pomona	1700	0	0	2200	2300	0	
SantaFe	0	0	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

;

(v) States with operating site:

(1 indicates the state will have at least one operating site)

Z [\*] :=

AZ	1
CA	1
NM	0
NV	0
UT	0

;

---

## 6.2 RESULTS (BASIC RECOMMENDATION)

Here we present the optimal (minimum-cost) expansion plan for BestChip to the western US:

- The cost of this expansion (minimum-cost) will be: \$ 1,440,480 per year

Based the current data, the recommended (minimum-cost) expansion plan involves opening two sites in Tucson, AZ and Pomona, CA.

Table 2 presents the amount of ingredients required at either of these new sites per year:

Table 2: Required ingredients at the new sites (ton/year)

	Corn	Wheat	Potato
Tucson, AZ	4720	3705	3475
Pomona, CA	5860	4645	4295

Table 3 presents the production plan of the new site:

Table 3: Production plan of the new sites (ton/year)

	Regular Chips	Green Onion Chips	Party Mix
Tucson, AZ	4100	2900	4900
Pomona, CA	5100	3500	6200

Table 4 to 6 present distribution of the products among the customers for each site:

Table 4: Distribution of Regular Chips among customers (ton/year)

	Jones <u>Demand =1300</u>	YZCO <u>Demand =1400</u>	Square Q <u>Demand =1200</u>	AJ Stores <u>Demand =1900</u>	Sun Quest <u>Demand =1900</u>	Harm's Path <u>Demand =1500</u>
Tucson, AZ	0	1400	1200	0	0	1500
Pomona, CA	1300	0	0	1900	1900	0

Table 5: Distribution of Green Onion Chips among customers (ton/year)

	Jones <u>Demand =900</u>	YZCO <u>Demand =1100</u>	Square Q <u>Demand =800</u>	AJ Stores <u>Demand =1200</u>	Sun Quest <u>Demand =1400</u>	Harm's Path <u>Demand =1000</u>
Tucson, AZ	0	1100	800	0	0	1000
Pomona, CA	900	0	0	1200	1400	0

Table 6: Distribution of Party Mix among customers (ton/year)

	Jones <u>Demand =1700</u>	YZCO <u>Demand =1700</u>	Square Q <u>Demand =1800</u>	AJ Stores <u>Demand =2200</u>	Sun Quest <u>Demand =2300</u>	Harm's Path <u>Demand =1400</u>
Tucson, AZ	0	1700	1800	0	0	1400
Pomona, CA	1700	0	0	2200	2300	0

## 7 QUESTION PART A

---

**Question:** If gasoline gets more expensive increasing the trucking cost for shipping the demand, then how is your recommendation affected?

A sensitivity analysis is conducted by incrementing the unit shipping cost of the products. To do this, the “basic” problem (that is, the problem with the current shipping unit cost of \$0.15 per mile) is solved, and its optimal solution is kept as the basic solution. Then, iteratively, the unit shipping cost is incremented and the problem is solved with this new value again. At each iteration, the new solution is compared to the basic solution to see if there is any change in the optimal solution. Clearly, it is sufficient to compare the operating sites in each iteration in that if the operating sites are not changed the production and distribution plans will remain unchanged. In fact, increasing the shipping cost may justify a higher fixed purchase cost to gain a lower total cost and so result in a different set of sites as the optimal expansion plan.

In order to compare the solutions, the following command is used:

```
if sum {i in SITE} (abs(Bsc_sol[i]-New_sol[i]))>=1 then {#report change of solution}
```

where `Bsc_sol` refers to the basic solution, and `New_sol` refers to the new solution, and `abs` is the Absolute Value operator. It is obvious that if values of  $W_i$  remain unchanged, the left-hand side of the above condition will be indeed zero. However, if any variable changes the left-hand side will become at least 1.

The increment step of \$0.001 (0.1 cent) per mile was chosen for this purpose. The results shows that the basic optimal solution is so sensitive to increase of shipping cost. In fact, the optimal solution changes in the seventh iteration (that is, less than 1 cent increase). If the cost of shipping the products increases to \$0.157 per mile the basic solution will not be optimal anymore. In the new setting, it is recommended to open a third site in addition to the two originally recommended ones. Clearly this changes the production and distribution plans of the former sites. More details about this revised plan is presented in the next section.

- The next question would be, how far would this new setting remain optimal?

The same analysis were conducted to find the next change of the optimal solution due to increase in unit shipping cost of the products. Results shows that even with huge increases (for example \$500 per mile) the optimal solution will not change anymore.

Next is a detailed recommendation based on the analysis results of this part:

## 7.1 RECOMMENDATION

Given an increase in the unit shipping cost of the products from the current rate of \$0.15 per mile, if the new cost is less than \$0.157 per mile (i.e. less than 4.6 % increase) the basic expansion plan, presented in the previous section, remains optimal. However, any changes further than this value changes the optimal solution of the problem, and necessitates revising the expansion plan as follows:

In this case, the recommended (minimum-cost) expansion plan involves opening three sites in Tucson, AZ, Pomona, CA, and Flagstaff, AZ.

Tables 7 to 11 present the revised production and distribution plans in this new setting:

Table 7: Required ingredients at the new sites (ton/year)

	Corn	Wheat	Potato
Tucson, AZ	3070	2410	2220
Pomona, CA	4340	3400	3160
Flagstaff, AZ	3170	2540	2390

Table 8: Production plan of the new sites (ton/year)

	Regular Chips	Green Onion Chips	Party Mix
Tucson, AZ	2700	1800	3200
Pomona, CA	3800	2600	4500
Flagstaff, AZ	2700	2000	3400

Table 9: Distribution of Regular Chips among customers (ton/year)

	Jones Demand =1300	YZCO Demand =1400	Square Q Demand =1200	AJ Stores Demand =1900	Sun Quest Demand =1900	Harm's Path Demand =1500
Tucson, AZ	0	0	1200	0	0	1500
Pomona, CA	0	0	0	1900	1900	0
Flagstaff, AZ	1300	1400	0	0	0	0

Table 10: Distribution of Green Onion Chips among customers (ton/year)

	Jones Demand =900	YZCO Demand =1100	Square Q Demand =800	AJ Stores Demand =1200	Sun Quest Demand =1400	Harm's Path Demand =1000
Tucson, AZ	0	0	800	0	0	1000
Pomona, CA	0	0	0	1200	1400	0
Flagstaff, AZ	900	1100	0	0	0	0

Table 11: Distribution of Party Mix among customers (ton/year)

	Jones Demand =1700	YZCO Demand =1700	Square Q Demand =1800	AJ Stores Demand =2200	Sun Quest Demand =2300	Harm's Path Demand =1400
Tucson, AZ	0	0	1800	0	0	1400
Pomona, CA	0	0	0	2200	2300	0
Flagstaff, AZ	1700	1700	0	0	0	0

---

## 7.2 AMPL RUN FILE

```
solve;
option display_eps 0.0001;
param Bsc_sol {SITE} >=0;      # The basic solution
param New_sol {SITE} >=0;      # New solution
let {i in SITE} Bsc_sol[i] := W[i];
for {1..1000} {
    let UnitShip := UnitShip + 0.001;
    solve;
    let {i in SITE} New_sol[i] := W[i];
    if sum {i in SITE} (abs(Bsc_sol[i] - New_sol[i])) >= 1 then {
        printf "\n" >> 'PartA.log';
        printf "Current unit shipping cost of products is: ($/ton.mile) \n" >>
            'PartA.log';
        display UnitShip >> 'PartA.log';
        printf "----- OPTIMAL OBJECTIVE VALUE ----- \n\n" >>
            'PartA.log';
        printf "The minimum cost of expansion is: ($/year)\n" >> 'PartA.log';
        display Total_Cost >> 'PartA.log';
        printf "\n" >> 'PartA.log';
        printf "----- OPTIMAL SOLUTION ----- \n\n" >>
            'PartA.log';
        printf "Here is details of the minimum-cost expansion plan:\n\n" >>
            'PartA.log';
        printf "(i) Operating sites: " >> 'PartA.log';
        printf "(1 indicates the site will be operating)\n" >> 'PartA.log';
        display W >> 'PartA.log';
        printf "\n" >> 'PartA.log';
        printf "(ii) Amount of ingredients to be shipped to each site\n" >>
            (ton/year):\n" >> 'PartA.log';
        display X >> 'PartA.log';
        printf "\n" >> 'PartA.log';
        printf "(iii) Distribution of products among the customers\n" >>
            (ton/year):\n" >> 'PartA.log';
        display Y >> 'PartA.log';
        printf "\n" >> 'PartA.log';
        printf "(iv) States with operating site:\n" >> 'PartA.log';
        printf "(1 indicates the state will have at least one operating site)\n" >>
            'PartA.log';
        display Z >> 'PartA.log';
        printf "\n" >> 'PartA.log';
        printf "----- \n" >>
            'PartA.log';
        quit;
    }
};
```

---



---

### 7.3 AMPL LOG FILE (OUTPUT)

Current unit shipping cost of products is: (\$/ton.mile)  
UnitShip = 0.157

----- OPTIMAL OBJECTIVE VALUE -----

The minimum cost of expansion is: (\$/year)  
Total\_Cost = 1480190

----- OPTIMAL SOLUTION -----

Here is details of the minimum-cost expansion plan:

(i) Operating sites: (1 indicates the site will be operating)

W [\*] :=

Yuma	0
Fresno	0
Tucson	1
Pomona	1
SantaFe	0
FlagStaff	1
LasVegas	0
StGeorge	0

;

(ii) Amount of ingredients to be shipped to each site (ton/year):

X [\*,\*] (tr)

:	CORN	WHEAT	POTATO	:=
Yuma	0	0	0	
Fresno	0	0	0	
Tucson	3070	2410	2220	
Pomona	4340	3400	3160	
SantaFe	0	0	0	
FlagStaff	3170	2540	2390	
LasVegas	0	0	0	
StGeorge	0	0	0	

;

(iii) Distribution of products among the customers (ton/year):

Y [Regular,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	0	1200	0	0	1500	
Pomona	0	0	0	1900	1900	0	
SantaFe	0	0	0	0	0	0	
FlagStaff	1300	1400	0	0	0	0	
LasVegas	0	0	0	0	0	0	

---

StGeorge	0	0	0	0	0	0
----------	---	---	---	---	---	---

[GrOnion,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath :=
Yuma	0	0	0	0	0	0
Fresno	0	0	0	0	0	0
Tucson	0	0	800	0	0	1000
Pomona	0	0	0	1200	1400	0
SantaFe	0	0	0	0	0	0
FlagStaff	900	1100	0	0	0	0
LasVegas	0	0	0	0	0	0
StGeorge	0	0	0	0	0	0

[PartyMix,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath :=
Yuma	0	0	0	0	0	0
Fresno	0	0	0	0	0	0
Tucson	0	0	1800	0	0	1400
Pomona	0	0	0	2200	2300	0
SantaFe	0	0	0	0	0	0
FlagStaff	1700	1700	0	0	0	0
LasVegas	0	0	0	0	0	0
StGeorge	0	0	0	0	0	0

;

(v) States with operating site:  
 (1 indicates the state will have at least one operating site)

Z [\*] :=

AZ	1
CA	1
NM	0
NV	0
UT	0

;

-----

## 8 QUESTION PART B

**Question:** If the rail freight cost for material shipping increases, then how is your recommendation affected?

As mentioned in the Assumptions section, it is assumed that shipping cost of all ingredients is increased at the same rate. A sensitivity analysis, similar to the one in the last part, is conducted. Ingredient shipping costs are incremented by 1% at each iteration, and the new solution is compared to the basic solution, using the same comparison condition, to identify the point where the optimal solution changes.

The analysis shows that the basic optimal solution remains valid unless the shipping cost of the ingredients increases by 891%. Given that this situation is very unlikely to happen in practice, we ignore the secondary question of “how far would this new setting remain optimal?”

Next is a new recommendation based on the analysis results of this part:

### 8.1 RECOMMENDATION

If the increase in shipping cost of the ingredients is less than 891%, the basic expansion plan remains optimal. However, any changes further than this value, which is very unlikely, changes the optimal solution of the problem, and necessitates revising the expansion plan as follows:

In this case, the recommended (minimum-cost) expansion plan involves opening two sites in Yuma, AZ, and Santa Fe, NM.

Tables 12 to 16 present the revised production and distribution plans in this new setting:

Table 12: Required ingredients at the new sites (ton/year)

	Corn	Wheat	Potato
Yuma, AZ	7690	6510	5800
Santa Fe, NM	2890	1840	1970

Table 13: Production plan of the new sites (ton/year)

	Regular Chips	Green Onion Chips	Party Mix
Yuma, AZ	6500	4400	9100
Santa Fe, NM	2700	2000	2000

Table 14: Distribution of Regular Chips among customers (ton/year)

	Jones Demand =1300	YZCO Demand =1400	Square Q Demand =1200	AJ Stores Demand =1900	Sun Quest Demand =1900	Harm's Path Demand =1500
Yuma, AZ	0	0	1200	1900	1900	1500
Santa Fe, NM	1300	1400	0	0	0	0

---

Table 15: Distribution of Green Onion Chips among customers (ton/year)

	Jones Demand =900	YZCO Demand =1100	Square Q Demand =800	AJ Stores Demand =1200	Sun Quest Demand =1400	Harm's Path Demand =1000
Yuma, AZ	0	0	800	1200	1400	1000
Santa Fe, NM	900	1100	0	0	0	0

Table 16: Distribution of Party Mix among customers (ton/year)

	Jones Demand =1700	YZCO Demand =1700	Square Q Demand =1800	AJ Stores Demand =2200	Sun Quest Demand =2300	Harm's Path Demand =1400
Yuma, AZ	1400	0	1800	2200	2300	1400
Santa Fe, NM	300	1700	0	0	0	0

---

## 8.2 AMPL RUN FILE

```
option solver cplex;
reset;
model basic.mod;
data basic.dat;
solve;
option display_eps 0.0001;
param Bsc_sol {SITE} >=0;           # The basic solution
param New_sol {SITE} >=0;          # New solution
let {i in SITE} Bsc_sol[i] := W[i];
param Inc {INGR,SITE} >= 0;
let {m in INGR , i in SITE} Inc[m,i] := 0.01 * SI[m,i];
# Increment of ingredients shipping cost equal to 1% of the basic shipping cost
for {1..1000} {
    let {m in INGR , i in SITE} SI[m,i] := SI[m,i] + Inc[m,i];
    solve;
    let {i in SITE} New_sol[i] := W[i];
    if sum {i in SITE} (abs(Bsc_sol[i] - New_sol[i])) >= 1 then {
        printf "\n" >> 'PartB.log';
        printf "Current shipping costs of the ingredients are: ($/ton) \n"
    >> 'PartB.log';
        display SI >> 'PartB.log';
        printf "----- OPTIMAL OBJECTIVE VALUE ----- \n\n" >>
        'PartB.log';
        printf "The minimum cost of expansion is: ($/year)\n" >> 'PartB.log';
        display Total_Cost >> 'PartB.log';
        printf "\n" >> 'PartB.log';
        printf "----- OPTIMAL SOLUTION ----- \n\n" >>
        'PartB.log';
        printf "Here is details of the minimum-cost expansion plan:\n\n" >>
        'PartB.log';
        printf "(i) Operating sites: " >> 'PartB.log';
        printf "(1 indicates the site will be operating)\n" >> 'PartB.log';
        display W >> 'PartB.log';
        printf "\n" >> 'PartB.log';
        printf "(ii) Amount of ingredients to be shipped to each site
        (ton/year):\n" >> 'PartB.log';
        display X >> 'PartB.log';
        printf "\n" >> 'PartB.log';
        printf "(iii) Distribution of products among the customers
        (ton/year):\n" >> 'PartB.log';
        display Y >> 'PartB.log';
        printf "\n" >> 'PartB.log';
        printf "(iv) States with operating site:\n" >> 'PartB.log';
        printf "(1 indicates the state will have at least one operating site)\n"
        >> 'PartB.log';
        display Z >> 'PartB.log';
        printf "\n" >> 'PartB.log';
        printf "----- \n" >>
        'PartB.log';
        quit;
    }
};
```

---

---

### 8.3 AMPL LOG FILE (OUTPUT)

Current shipping costs of the ingredients are: (\$/ton)

SI [\*,\*] (tr)

:	CORN	WHEAT	POTATO	:=
Yuma	89.1	44.55	142.56	
Fresno	106.92	71.28	98.01	
Tucson	80.19	89.1	133.65	
Pomona	98.01	62.37	124.74	
SantaFe	71.28	124.74	89.1	
FlagStaff	89.1	106.92	98.01	
LasVegas	115.83	106.92	80.19	
StGeorge	124.74	133.65	71.28	

;

----- OPTIMAL OBJECTIVE VALUE -----

The minimum cost of expansion is: (\$/year)

Total\_Cost = 3724430

----- OPTIMAL SOLUTION -----

Here is details of the minimum-cost expansion plan:

(i) Operating sites: (1 indicates the site will be operating)

W [\*,\*] :=

Yuma	1
Fresno	0
Tucson	0
Pomona	0
SantaFe	1
FlagStaff	0
LasVegas	0
StGeorge	0

;

(ii) Amount of ingredients to be shipped to each site (ton/year):

X [\*,\*] (tr)

:	CORN	WHEAT	POTATO	:=
Yuma	7690	6510	5800	
Fresno	0	0	0	
Tucson	0	0	0	
Pomona	0	0	0	
SantaFe	2890	1840	1970	
FlagStaff	0	0	0	
LasVegas	0	0	0	
StGeorge	0	0	0	

;

(iii) Distribution of products among the customers (ton/year):

---

Y [Regular,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	1200	1900	1900	1500	
Fresno	0	0	0	0	0	0	
Tucson	0	0	0	0	0	0	
Pomona	0	0	0	0	0	0	
SantaFe	1300	1400	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

[GrOnion,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	800	1200	1400	1000	
Fresno	0	0	0	0	0	0	
Tucson	0	0	0	0	0	0	
Pomona	0	0	0	0	0	0	
SantaFe	900	1100	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

[PartyMix,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	1400	0	1800	2200	2300	1400	
Fresno	0	0	0	0	0	0	
Tucson	0	0	0	0	0	0	
Pomona	0	0	0	0	0	0	
SantaFe	300	1700	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	0	0	0	0	0	0	

;

(v) States with operating site:

(1 indicates the state will have at least one operating site)

Z [\*] :=

AZ	1
CA	0
NM	1
NV	0
UT	0

;

## 9 QUESTION PART C

---

**Question:** How much time does CPLEX take to solve your linear MBP problem?

In order to answer this question, the following command is used after solving the problem:

```
display _total_solve_time;
```

It took seconds 0.03125 seconds to solve this problem on a system with Intel® Core™ i7-3520M 2.90 GHz CPU with 8.00 GB memory.

```
_total_solve_time = 0.03125
```



## 10 QUESTION PART D

---

**Question:** Solve the linear programming relaxation of your model. How much time does that take? How much difference do you notice between the objective value of the LP relaxation and your original MBP? Does your LP relaxation give a feasible solution to the MBP?

There are two ways to relax the MBP model. First is to use the following general command in the Run file before solving the problem:

```
model basic.mod;  
data basic.dat;  
option relax_integrality 1;  
solve;
```

The second is to relax the integrality constraints manually. This model has two sets of integer (binary) variables, namely  $W_i$  and  $Z_s$ . We need to eliminate the integrality requirement but keep their bounds. Thus declaration of these variable in the model file is revised as follows:

```
var W {SITE} >=0 , <=1;  
var Z {STAT} >=0 , <=1;
```

Two methods are identical and generate the same result.

Next is a solution to the LP relaxation model:

---

## 10.1 AMPL LOG FILE (OUTPUT)

----- OPTIMAL OBJECTIVE VALUE -----

The minimum cost of expansion is: (\$/year)  
Total\_Cost = 697990

----- OPTIMAL SOLUTION -----

Here is details of the minimum-cost expansion plan:

(i) Operating sites: (1 indicates the site will be operating)

W [\*] :=

Yuma	0
Fresno	0
Tucson	0.0077
Pomona	0.0109
SantaFe	0.0042
FlagStaff	0
LasVegas	0
StGeorge	0.0039
;	

(ii) Amount of ingredients to be shipped to each site (ton/year):

X [\*,\*] (tr)

:	CORN	WHEAT	POTATO	:=
Yuma	0	0	0	
Fresno	0	0	0	
Tucson	3070	2410	2220	
Pomona	4340	3400	3160	
SantaFe	1650	1295	1255	
FlagStaff	0	0	0	
LasVegas	0	0	0	
StGeorge	1520	1245	1135	
;				

(iii) Production plan for each site (ton/year):

Production [\*,\*]

:	Regular	GrOnion	PartyMix	:=
Yuma	0	0	0	
Fresno	0	0	0	
Tucson	2700	1800	3200	
Pomona	3800	2600	4500	
SantaFe	1400	1100	1700	
FlagStaff	0	0	0	
LasVegas	0	0	0	
StGeorge	1300	900	1700	
;				

---

(iv) Distribution of products among the customers (ton/year):

Y [Regular,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	0	1200	0	0	1500	
Pomona	0	0	0	1900	1900	0	
SantaFe	0	1400	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	1300	0	0	0	0	0	

[GrOnion,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	0	800	0	0	1000	
Pomona	0	0	0	1200	1400	0	
SantaFe	0	1100	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	900	0	0	0	0	0	

[PartyMix,\*,\*]

:	Jones	YZCO	SquareQ	AJStores	SunQuest	HarmsPath	:=
Yuma	0	0	0	0	0	0	
Fresno	0	0	0	0	0	0	
Tucson	0	0	1800	0	0	1400	
Pomona	0	0	0	2200	2300	0	
SantaFe	0	1700	0	0	0	0	
FlagStaff	0	0	0	0	0	0	
LasVegas	0	0	0	0	0	0	
StGeorge	1700	0	0	0	0	0	

;

(v) States with operating site:

(1 indicates the state will have at least one operating site)

Z [\*] :=

AZ	7.7e-09
CA	1.09e-08
NM	4.2e-09
NV	0
UT	3.9e-09

;

- How much time does that [solving the relaxed model] take?

`_total_solve_time = 0.015625` (sec)

- How much difference do you notice between the objective value of the LP relaxation and your original MBP?

Objective value of the original MBP problem	\$ 1,440,480	per year
Objective value of the LP Relaxation	\$ 697,990	per year

Objective value of the LP relaxation is \$ 742,490 (per year) less than the original problem. Actually LP relaxation's objective value is less than one half of the original MBP problem.

- Does your LP relaxation give a feasible solution to the MBP?

No;

In the LP relaxation, “site operation indicator” variables get fractional values, which make the LP relaxation solution infeasible to the original problem. In fact, such fractional values are meaningless to the context of the original problem. The same applies to “state operation indicator” variables.